The Ising Model

A one-dimensional Ising model. See *Thermal Physics*, by Daniel V. Schroeder, pg. 349.

In[68]:=
```
Clear["Global`*"]
```

In[69]:=
```
initialize[L_] := Table[ RandomChoice[{-1, 1}], { x, 1, 1}, {y, 1, L}];
```

In[70]:=
```
L = 100;
Nsteps = 20 * L * L; (* Guess for how many iterations to run *)
s = initialize[L];
```

In[72]:=
```
ΔU[i_] := Module[{left, right},
  left = If[i > 1, i - 1, L];
  right = If[i < L, i + 1, 1];
  (* Energy of flipping spin i *)
  2 * s〚1, i〛 * (s〚1, left〛 + s〚1, right〛)
 ]
e[s_] := -Sum[s〚1, i〛 s〚1, i + 1〛, {i, 1, L - 1}] (* Total energy *)
etheory[T_] := -L Tanh[1 / T]
```

In[75]:=
```
newrun[T_, Nsteps_] := Module[{},
  initialize[L];
  run[T, Nsteps] // Quiet
 ]
run[T_, Nsteps_] := Module[{},
  Do[i = RandomInteger[{1, L}];
   Ediff = ΔU[i];
   If[Ediff ≤ 0,
    s〚1, i〛 = - s〚1, i〛;,
    If[RandomReal[1.0] < Exp[-Ediff / T],
     s〚1, i〛 = -s〚1, i〛;
    ]
   ];, {n, 1, Nsteps}];
  e[s]
 ]
show[s_, T_] := MatrixPlot[s, AspectRatio → 1 / 10,
  MaxPlotPoints → Infinity,  PlotLabel → T, ImageSize → Large]
showSmall[s_] := MatrixPlot[s, AspectRatio → 1 / 10,
  MaxPlotPoints → Infinity, Axes → False, ImageSize → Large, Frame → False]
```

In[79]:=
```
Dynamic[show[s, T]]
```

Out[79]=



First, just show some relaxation at low temperature. Then Look at high temperature.

In[80]:=
```
s = initialize[L];
```

In[81]:=
```
run[1.49, 100 000] (* run[T, Nsteps].  Return value is the energy. *)
```

Out[81]=

-63

Will it ever settle down to a fully-magnetized state?

In[82]:=
```
out = Table[{T, newrun[T, Nsteps], showSmall[s]}, {T, 0.1, 2, 0.1}];
EvsT = out⟦All, {1, 2}⟧;
TableForm[out, TableHeadings → {None, {"T", "Energy", "State"}}]
ListPlot[EvsT, LabelStyle → Larger,
 AxesLabel → {"T", "Total Energy"}, ImageSize → Scaled[0.8]]
```

Out[84]//TableForm=

T	Energy	State
0.1	-99	
0.2	-99	
0.3	-99	
0.4	-99	
0.5	-99	
0.6	-93	

0.7	−91	
0.8	−71	
0.9	−79	
1.	−79	
1.1	−67	
1.2	−39	
1.3	−75	
1.4	−63	
1.5	−67	
1.6	−61	
1.7	−45	
1.8	−51	
1.9	−47	

2. −35 

Out[85]=



What is the equilibrium at T = 0.5? Do we need to run even longer? You can try it, but it’s difficult to know when everything has converged! Instead, do an analytic calculation for the expected energy. (See Eq. 8.44 in Schroeder.) It undergoes a gradual continuous transition down to fully magnetized.

In[86]:=
```
{Tmin, Tmax} = MinMax[EvsT⟦All, 1⟧];
Show[{
  Plot[etheory[T], {T, Tmin, Tmax}, PlotStyle → Red], ListPlot[EvsT]
 }, LabelStyle → Larger, AxesLabel → {"T", "E"}, ImageSize → Large]
```

Out[87]=